/*
Matthew Baum	
Title: Crime Reporting and Adjudication in US Rape Culture

	
HYPOTHESES
	
Stated-Hyp1: Cases involving victims of color are less likely to be punished than 
	cases involving white victims, all else equal.
	
	*Test-Hyp1: In the punish condition, respondents will be more likely to recommend cases involving white victims vs. non-white [all else is already equal because respondents are randomized into experimental conditions].

	
	
Stated-Hyp2: Cases involving female victims with a promiscuous sexual history (i.e. 
	multiple sexual partners) are less likely to be recommended for 
	reporting/punishment than cases involving single or married victims, all
	else equal.
	
	*Test-Hyp2: In the reporting condition, respondents will be less likely to recommend  cases involving women victims with multiple partners vs. those who did not have multiple partners (i.e. single or married combined)

	*Test-Hyp3: In the punish condition, respondents will be less likely to recommend  cases involving women victims with multiple partners vs. those who did not have multiple partners (i.e. single or married combined)	
	
	
	
Stated-Hyp3: Cases involving female victims who were dressed provocatively (i.e. 
	night club outfit) at the time of the incident are less likely to be 
	recommended for reporting or punishment, all else equal.
	
	*Test-Hyp4: In the reporting condition, respondents will be less likely to recommend  cases involving women victims who were dressed provocatively vs. those in work attire.

	*Test-Hyp5: In the punish condition, respondents will be less likely to recommend  cases involving women victims  who were dressed provocatively vs. those in work attire.
	
	
	
	
Stated-Hyp4: Cases involving white defendants are less likely to be recommended for 
	reporting/punishment than defendants of color, all else equal.
	
	*Test-Hyp6: In the reporting condition, respondents will be less likely to recommend  cases involving white defendants vs. defendants of color

	*Test-Hyp7: In the punish condition, respondents will be less likely to recommend  cases involving white defendants vs. defendants of color
	
	
	
Stated-Hyp5: Cases involving defendants with a high socio-economic status are less 
	likely to be recommended for reporting/punished for rape than defendants 
	with a low socio-economic status, all else equal.	
	
	*Test-Hyp8: In the reporting condition, SES of the perpetrator will be negatively associated with liklihood of being recommended. 

	*Test-Hyp9: In the punish condition, SES of the perpetrator will be negatively associated with liklihood of being recommended. 
	
	
	
Stated-Hyp6: Cases involving victims with an existing felony record are less likely 
	to be recommended for reporting or severe punishment, all else equal.
	
	*Test-Hyp10: In the reporting condition, respondents will be less likely to recommend  cases involving victims with a felony record vs. victims without a felony record

	*Test-Hyp11: In the punish condition, respondents will be less likely to recommend  cases involving victims with a felony record vs. victims without a felony record
	
	
	
Stated-Hyp7: Rape incidents that took place at a party are less likely to be 
	recommended for reporting or punishment than rape cases that took place 
	at the victim’s home or in a park, all else equal.

	*Test-Hyp12: In the reporting condition, respondents will be less likely to recommend  cases that took place at a party vs. not at a party (victim's home or park)

	*Test-Hyp13: In the punish condition, respondents will be less likely to recommend  cases that took place at a party vs. not at a party (victim's home or park)	
	
	
Stated-Hyp8: Rape by an acquaintance is less likely to be recommended for reporting 
	or severe punishment than stranger rape, all else equal.
	
	*Test-Hyp14: In the reporting condition, respondents will be less likely to recommend  cases that were committeed by an acquaintance vs. a stranger

	*Test-Hyp15: In the punish condition, respondents will be less likely to recommend  cases that that were committeed by an acquaintance vs. a stranger
	
	
	
Stated-Hyp9: Cases involving male victims with male perpetrators are less likely to 
	be recommended for reporting or punishment than cases involving female 
	victims with male perpetrators, all else equal.
	
	*Test-Hyp16: In the reporting condition, respondents will be more likely to recommend  cases that involved male perpetrators+female victims vs. male perpetrators+male victims 

	*Test-Hyp17: In the punish condition, respondents will be more likely to recommend  cases that that involved male perpetrators+female victims vs. male perpetrators+male victims 
	
********************************************************************************	
NOTES:
- Hypotheses are in appendix 2 (pages 13-14 in the proposal).

- there are a couple of phantom conditions (e.g. location entails only one 
comparison; home and park need to be combined; sexual history also has just one 
comparison but 3 experimental groups)

- SES is a 3-category variable. Based on table in page 3 of the proposal, it 
appears that the businessman is supposed to have high SES, athlete is in the 
middle, and construction worker has lowest SES. 

- perpetrator is always a man. so last two hypotheses are just testing gender 
of victim.

*/

clear all
import delimited Baum_TESS_Data_Long_Clean.csv, clear


********************************************************************************

* CONSTRUCT INDICATORS OF EXPERIMENTAL MANIPULATIONS

	* whether R was selected into the report or punish condition
	tab report punish 
	// absolutely orthogonal; so these are likely experimental conditions	

	* victim's race
		tab vic_black vic_white, mis
	
	* victim's gender 
		tab vic_sex
		gen vic_female=1 if vic_sex=="Female"
		replace vic_female=0 if vic_sex=="Male"
		tab vic_female
			
	* type of crime: rape vs. robbery
		tab rape robbery, mis

	* victim's sexual history (multiple partners vs. married)
		lab def sexhist 1 "single" 2 "married" 3 "multiple partners"
		gen vic_sexhistory=1 if vic_single==1
		replace vic_sexhistory=2 if vic_married==1
		replace vic_sexhistory=3 if vic_partners==1
		lab val vic_sexhistory sexhist
		tab vic_sexhistory, mis
		
		recode vic_sexhistory (1/2=0) (3=1), gen(vic_multiplepartners) 
		
	* victim's outfit (provocative)
		gen vic_provoutfit=1 if outfit== "Night Club Attire"
		replace vic_provoutfit=0 if outfit=="Work Attire"
		tab vic_provoutfit, mis

	* defendent race (white vs. all others)
		tab perp_black perp_white, mis

	* defendent SES
		/* I am assuming that perpetrators' profession is the manipulation for ses
		and that businessmen are supposed to be higher ses than athletes
		*/
		tab perp_prof, mis
		lab def ses 1 "Worker" 2 "Athlete" 3 "Business man"
		gen perp_ses=1 if perp_prof=="Worker"
		replace perp_ses=2 if perp_prof=="Athlete"
		replace perp_ses=3 if perp_prof=="Business man"
		lab val perp_ses ses
		tab perp_ses, mis

	* perpetrator relation: acquaintance
		tab perp_stranger perp_acquain, mis
		
	* victim's history of felony	
		tab vic_criminal, mis
		gen vic_felon = 1 if vic_criminal =="Minor felony record"
		replace vic_felon=0 if vic_criminal=="No criminal record"
		tab vic_felon, mis

	* rape location
		tab location
		gen location_party= 0 if location == "Home"
		replace location_party= 0 if location == "Park"		
		replace location_party= 1 if location == "Party"	
		tab location_party
		
* CONSTRUCT OUTCOME MEASURES

* whether respondent chose to report or punish
	tab chosen
	* note: we don't have full documentation on this. But we are assuming that chosen means that this vignette was chosen
	// I'm not sure why the mean is not 0.5; but I randomly checked within individual means and all the ones I checked were 0.5. 
	
********************************************************************************

* ANALYSIS
	
*Test-Hyp1: In the punish condition, respondents will be more likely to recommend cases involving white victims vs. non-white [all else is already equal because respondents are randomized into experimental conditions].

	reg chosen i.vic_white if punish==1, vce(cluster id)
	// reject. 0.003W
	
	tess 1.vic_white +, init(Baum871)
		
	
*Test-Hyp2: In the reporting condition, respondents will be less likely to recommend  cases involving women victims with multiple partners vs. those who did not have multiple partners (i.e. single or married combined)

	reg chosen i.vic_multiplepartners if report==1& vic_female==1, vce(cluster id)
	// do not reject. 0.000
	
	tess 1.vic_multiplepartners -
	
*Test-Hyp3: In the punish condition, respondents will be less likely to recommend  cases involving women victims with multiple partners vs. those who did not have multiple partners (i.e. single or married combined)	
	
	reg chosen i.vic_multiplepartners if punish==1& vic_female==1, vce(cluster id)
	// do not reject. 0.000
	tess 1.vic_multiplepartners -
		

*Test-Hyp4: In the reporting condition, respondents will be less likely to recommend  cases involving women victims who were dressed provocatively vs. those in work attire.

	reg chosen i.vic_provoutfit if report==1& vic_female==1, vce(cluster id)
	// do not reject. 0.000
	tess 1.vic_provoutfit -
	
	
	
*Test-Hyp5: In the punish condition, respondents will be less likely to recommend  cases involving women victims  who were dressed provocatively vs. those in work attire.
	
	reg chosen i.vic_provoutfit if punish==1& vic_female==1, vce(cluster id)
	// reject. 0.097
	tess 1.vic_provoutfit -
		
	
*Test-Hyp6: In the reporting condition, respondents will be less likely to recommend  cases involving white defendants vs. defendants of color

	reg chosen i.perp_white if report==1, vce(cluster id)
	// reject. 0.000W
	tess 1.perp_white -
		

*Test-Hyp7: In the punish condition, respondents will be less likely to recommend  cases involving white defendants vs. defendants of color
	
	reg chosen i.perp_white if punish==1, vce(cluster id)
	// reject. 0.000W	
	tess 1.perp_white -	

*Test-Hyp8: In the reporting condition, SES of the perpetrator will be negatively associated with liklihood of being recommended. 

	reg chosen perp_ses if report==1, vce(cluster id)
	// reject. 0.518
	tess perp_ses -

*Test-Hyp9: In the punish condition, SES of the perpetrator will be negatively associated with liklihood of being recommended. 
	
	reg chosen perp_ses if punish==1, vce(cluster id)
	// reject.  0.953
	tess perp_ses -	
	
*Test-Hyp10: In the reporting condition, respondents will be less likely to recommend cases involving victims with a felony record vs. victims without a felony record

	reg chosen i.vic_felon if report==1, vce(cluster id)
	// do not reject. 0.000
	tess 1.vic_felon -

*Test-Hyp11: In the punish condition, respondents will be less likely to recommend cases involving victims with a felony record vs. victims without a felony record
	
	reg chosen i.vic_felon if punish==1, vce(cluster id)
	// reject. 0.270
	tess 1.vic_felon -
	
*Test-Hyp12: In the reporting condition, respondents will be less likely to recommend  cases that took place at a party vs. not at a party (victim's home or park)

	reg chosen i.location_party if report==1, vce(cluster id)
	// do not reject. 0.000
	tess 1.location_party -
	
*Test-Hyp13: In the punish condition, respondents will be less likely to recommend  cases that took place at a party vs. not at a party (victim's home or park)	
	
	reg chosen i.location_party if punish==1, vce(cluster id)
	// do not reject. 0.000
	tess 1.location_party -	

*Test-Hyp14: In the reporting condition, respondents will be less likely to recommend  cases that were committeed by an acquaintance vs. a stranger

	reg chosen i.perp_acquain if report==1, vce(cluster id)
	// do not reject. 0.000
	tess 1.perp_acquain -
	
*Test-Hyp15: In the punish condition, respondents will be less likely to recommend  cases that that were committeed by an acquaintance vs. a stranger
	
	reg chosen i.perp_acquain if punish==1, vce(cluster id)
	// do not reject. 0.000	
	tess 1.perp_acquain -

*Test-Hyp16: In the reporting condition, respondents will be more likely to recommend  cases that involved male perpetrators+female victims vs. male perpetrators+male victims 

	reg chosen i.vic_female if report==1, vce(cluster id)
	// do not reject. 0.000
	tess 1.vic_female +

*Test-Hyp17: In the punish condition, respondents will be more likely to recommend  cases that that involved male perpetrators+female victims vs. male perpetrators+male victims 

	reg chosen i.vic_female if punish==1, vce(cluster id)
	// do not reject. 0.000
	tess 1.vic_female +	
	
